



UNIVERSIDADE DA CORUÑA

FACULDADE DE INFORMÁTICA Departamento de Computación Estructura de Datos e da Información

Práctica 0c: Recursividad + Punteros

Esta práctica ilustra la implementación de una lista enlazada de enteros. El programa hace uso de una `unit` que agrupa operaciones básicas de manejo de una lista: `CrearLista`, `Insertar`, `Borrar` e `Imprimir`, alguna de ellas implementada de manera recursiva. Situar la `unit` en un fichero de nombre `lista.pas` y compilarlo. Hacer lo mismo con el programa principal para observar cómo puede utilizarse una `unit`.

```
unit lista;

Interface

type
  tPos = ^Nodo;
  Nodo = record
    numero: integer;
    sig: tPos;
  end;
  tLista = tPos;

procedure CrearLista (var L: tLista);
procedure Insertar(var L: tLista; x: integer);
procedure Imprimir (L: tLista);
procedure BorrarLista (var L: tLista);

Implementation

procedure CrearLista (var L: tLista);
  {Inicializar el puntero de la lista}
begin
  L:= nil;
end;

procedure Insertar(var L: tLista; x: integer);
  {Insertar un nuevo elemento en la cabeza de una lista }
var
  nuevo: tPos;
begin
  new(nuevo);          {Crea una variable apuntada (registro)}
  nuevo^.numero:= x;  {Da valores a los campos: un entero y }
  nuevo^.sig:= L;     {un puntero a otro registro (primero) }
  L:= nuevo;          {El nuevo elemento es cabeza de lista }
end;

procedure Imprimir (L: tLista);
  {Recorrer los elementos de la lista para imprimirlos}
var
```

```

    temp: tPos;
begin
    if L <> nil then
        begin
            write('Lista de numeros: ');
            temp:= L;
            while (temp<>nil) do begin
                write (temp^.numero, '->');
                temp:= temp^.sig;
            end;
            writeln;
        end else writeln ('Lista vacia');
end;

procedure BorrarLista (var L: tLista);
{Borrar todos los elemento de la lista}
var
    temp: tPos;
begin
    if L<>nil
    then begin
        temp:= L;
        L:= L^.sig;
        writeln ('Borrando ->', temp^.numero);
        dispose(temp);
        BorrarLista(L);
    end;
end;

begin
end.

program plista;
uses lista;

var
    L : tLista;
    n: integer;
    fin: boolean;

begin
    CrearLista(L);
    fin:= false;
    writeln ('Programa para crear una lista de numeros');
    repeat
        write ('Introducir un numero (0 para terminar) '); readln(n);
        if (n<>0)
        then Insertar(L,n)
        else fin:= true;
    until fin;
    writeln;
    Imprimir (L);
    writeln;
    BorrarLista (L);
    writeln;
    Imprimir (L);
end.

```